



FlexPDE の基本機能

デモ版は製品として 30 日間動作します。最初に FlexPDE の基本的な機能と製品コンセプトをご理解ください。

動作確認

製品のインストールし、デモ版のライセンスを取得したら、まずは動作確認を行います。FlexPDE を起動し、次のように操作します。

1. File メニューから Open File コマンドを利用して
C:\Program Files\FlexPDE8\Samples\Usage\Plotting\3d_ploton.pde
を開きます。
2. Control メニューから Run Script を選択するか、または、 ボタンをクリックします。
3. 計算終了時の様子を次にします。



FlexPDE の計算実行後の画面


このような画面が表示されたら、FlexPDE は正常に動作しています。次にこの画面を使ってメニュー構造と画面の切替え方法を説明します。

エディタと操作画面

スクリプトの計算が完了すると、画面上にスクリプトの PLOT セクションで指定した画像を表示します。スクリプトの PLOT セクションコマンドを次に示します。


1. `grid(x,y,z) on region 1`
2. `grid(y,z) on x=0 on 'plug'`
3. `contour(u) on x=0.51 on layer 2`
4. `contour(u) on z=0.51 on region 2`
5. `contour(u) on surface 2 on region 2 { on paraboloidal layer interface }`
6. `vector(grad(u)) on surface 2 on 'outer'`

特定のウィンドウを全画面表示する場合は、目的のウィンドウをダブルクリックします。




(No.1) `grid(x,y,z) on region 1`


全画面表示されたウィンドウから元のマルチウィンドウに戻る場合は、ウィンドウでマウスの右ボタンをクリックし、ポップアップメニューで Restore を選択します。




(No.2) grid(y,z) on x=0 on 'plug'




(No.3) contour(u) on x=0.51 on layer 2



(No.4) contour(u) on z=0.51 on region 2




(No.5)contour(u) on surface 2 on region 2 { on paraboloidal layer interface }





(No.6) vector($\text{grad}(u)$) on surface 2 on 'outer'

プロット画面の左側にあるメッシュの様子を示すウィンドウをメッシュウィンドウと呼びます。これをダブルクリックすると全画面表示します。



メッシュウィンドウ

全画面表示したメッシュウィンドウからマルチ画面にモデル場合も Restore コマンドを利用します。

スクリプトの画面に戻る場合は、メニュー操作で Control:Show Editor とするか、または、Show Editor ボタン  をクリックします。スクリプトを編集したら、再度、計算を実行します。再計算せずに、再びプロット画面を表示する時は Control:Show Plots とするか、または Show Plots ボタン  をクリックします。

スクリプトの概略

FlexPDE ではスクリプトを使って偏微分方程式 (常微分方程式も含む) や境界条件などを記述します。スクリプトはいくつかのセクションで構成されます。次の手順にしたがって、新しいスクリプト作成画面を表示します。

1. File メニューから New Script を選択します。
2. ダイアログに新しいスクリプトファイルの名前 (ここでは test) を入力し、保存するフォルダを決めて OK ボタンをクリックします。
3. 次に示すテンプレート画面を表示します。

test.pde の内容

Fill in the following sections (removing comment marks ! if necessary), and delete those that are unused.}

```
TITLE 'New Problem' { the problem identification }
COORDINATES cartesian2 { coordinate system, 1D,2D,3D, etc }
VARIABLES { system variables }
  u { choose your own names }
! SELECT { method controls }
! DEFINITIONS { parameter definitions }
! INITIAL VALUES
EQUATIONS { PDE's, one for each variable }
  div(grad(u))=0 { one possibility }
! CONSTRAINTS { Integral constraints }
BOUNDARIES { The domain definition }
  REGION 1 { For each material region }
  START(0,0) { Walk the domain boundary }
  LINE TO (1,0) TO (1,1) TO (0,1) TO CLOSE
! TIME 0 TO 1 { if time dependent }
MONITORS { show progress }
PLOTS { save result displays }
  CONTOUR(u)
END
```

重要: スクリプト本文はもちろん、コメント部分にも日本語は利用できません。編集画面上でキーボードから日本語を入力することは可能ですが、次にファイルを開くと文字化けして判別不能になります。

重要: コメント文の先頭には感嘆符「!」を付けます。

スクリプト

スクリプトは上記に示すように主に、COORDINATES, VARIABLES, EQUATIONS, BOUNDARIES, MONITORS, PLOTS セクションで構成されます。ここではそれらのセクションについて簡単に説明します。

COORDINATES このセクションは次の要領で記述します。

```
COORDINATES geometry
```

geometry には次のものを利用します。

名前	説明
CARTESIAN1	1次元座標 'X'
CYLINDER1	1次元円柱座標 'R'
SPHERE1	1次元球座標 'R'
CARTESIAN2	2次元座標 'X' と 'Y'
XCYLINDER	2次元円柱座標。'Z' は水平 X 軸, 半径 'R' は Y 軸に沿っています。
YCYLINDER	2次元円柱座標。'Z' は水平 Y 軸, 半径 'R' は X 軸に沿っています。
CARTESIAN3	1次元直交座標 'X', 'Y', 'Z'.

ここでは,

```
COORDINATES cartesian2 { coordinate system, 1D,2D,3D, etc }
```

であり, 2次元座標を利用していることが分かります。

VARIABLES このセクションで定義する変数は有限要素法を利用して求めるものとしします。また, 各変数は問題の定義域において連続なスカラーフィールドを定義するものとしします。変数は EQUATIONS セクションに記述する偏微分方程式でも利用することになります。

従属変数の命名法は次の規則に従います。

- 変数名は半角のアルファベットとします。先頭に数字や記号は利用できません。
- 変数名として 1 文字だけの t は利用できません。t は時間変数としての予約語です。
- 変数名の文字数に制限はありません。文字, 数字, 記号を組み合わせ利用できます。
- 変数名でセパレータは利用できません。例えば, temperature.celsius のような結合型の変数を利用する場合は '.' を使います。
- 変数名に '-' (マイナス) 記号は利用できません。これは予約語になっています。

ここでは,

```
VARIABLES { system variables }  
u { choose your own names }
```

従属変数名を u とします。

EQUATIONS 問題の従属変数を定義する偏微分方程式は EQUATIONS セクションに記述します。VARIABLES セクションで定義した従属変数は必ずどこかの方程式に含まれます。ディスクリプタに数式を記述する際は、ノートに書き込む要領で行います。演算子としては発散 DIV, 勾配 GRAD, 回転 CURL, ラプラシアン DEL2 などを利用してください。FlexPDE は COORDINATES セクションで選択した座標系でこれらの演算子を展開します。

```
EQUATIONS { PDE's, one for each variable }
  div(grad(u))=0 { one possibility }
```

式 $\text{div}(\text{grad}(u))=0$ は次に示すラプラス方程式です。

$$\nabla^2 U \equiv \nabla \cdot \nabla U \equiv \text{div}(\text{grad}(U)) = \frac{\partial^2 U}{\partial x^2} + \frac{\partial^2 U}{\partial y^2} = 0$$

BOUNDARIES 目的のシステムが存在する問題の定義域と、その定義域を囲む境界条件を記述します。FlexPDE は歴史的に 2次元ベースの境界条件を使って問題を解くスタイルを採用してきました。3次元問題は2次元の定義域を3次元方向に拡張することによって対応します。1次元定義域は2次元の応用として処理しています。すべてのモデルディスクリプタには必ず BOUNDARIES セクションを作成します。境界は各材料領域の外周のことです。2次元直交空間ではその外周を境界パスと呼びます。定義域は物理的に REGION, FEATURE, EXCLUDE のサブセクションに分けることができます。ディスクリプタには少なくとも1つの REGION サブセクションが存在します。FEATURE と EXCLUDE はオプションです。すべてのサンプルファイルで REGION を利用していますのでご参照ください。

```
BOUNDARIES { The domain definition }
  REGION 1 { For each material region }
  START(0,0) { Walk the domain boundary }
  LINE TO (1,0) TO (1,1) TO (0,1) TO CLOSE
```

境界領域には領域を示す REGION や LAYER があります。利用した座標軸における物理的な範囲を座標点で設定します。

MONITORS, PLOTS 問題の計算途中のグラフィックスをリスト形式で表示する場合は MONITORS セクションを利用します。一方、問題やステージの計算完了時、または指定した時刻における画像をリスト形式で表示する場合は PLOT を利用します。PLOTS および MONITORS の文型は同じです。

```
MONITORS { show progress }
PLOTS { save result displays }
  CONTOUR(u)
```

計算中の経過を MONITORS で、計算完了時の様子を PLOT で表示します。このサンプルでは平面上にメッシュを作成するだけなので、MONITORS の効果はありません。

スクリプトを作成する

実際に簡単なスクリプトを作成してみましょう。先のセクションで利用した test.pde(p.6) を使って、新たなモデルを作成します。目的のスクリプトを次に示します。コメント部分にここでは日本語を入力してい

ますが、プログラム中には入力しないでください。

```
TITLE 'Heat flow around an Insulating blob' { タイトルを変更します }
COORDINATES cartesian2
VARIABLES
  phi { 変数名を phi に変更します }
DEFINITIONS { 定数を定義します }
  k=1
  r=0.5
EQUATIONS
  div(-k*grad(phi))=0 { マイナス記号を付けます }
BOUNDARIES { 境界条件を定義します }
  REGION 1 'box' { 全領域 }
    START(-1,-1) { 正方形の 4 つの境界条件を順番に設定します }
    VALUE(phi)=0 LINE TO (1,-1) { 第 1 種の境界条件 }
    NATURAL(phi)=0 LINE TO (1,1) { 第 2 種の境界条件 }
    VALUE(phi)=1 LINE TO (-1,1)
    NATURAL(phi)=0 LINE TO CLOSE
  REGION 2 'blob' { 中央の円の領域 }
    k=0.001 { 定数を変更します }
    START 'ring' (r,0)
    ARC(CENTER=0,0) ANGLE=360 TO CLOSE
PLOTS
  CONTOUR(phi) { 図 1. 温度の等高線図 }
  VECTOR(-k*grad(phi)) { 図 2. 流束のベクトル図 }
  ELEVATION(Phi) FROM (0,-1) to (0,1){ 図 3. 縦の中央線上の温度変化 }
  ELEVATION(Normal(-k*grad(phi))) ON 'ring' { 図 4. 円周上の流束 }
END
```

スクリプトを入力したらそれを上書き保存します。スクリプトを実行した時のプロットの様子を次にします。




図 1. 温度の等高線図




図 2. 流束のベクトル図




図 3. 縦の中央線上の温度変化




図 4. 円周上の流束

スクリプトで利用したいいくつかのコマンドについて説明します。

VALUE 第 1 種の境界値問題で利用します。つまり、空間のある与えられた領域で偏微分方程式が成り立ち、その領域の境界上での解の値が指定される場合に利用します。

NATURAL 第 2 種の境界値問題で利用します。空間のある領域で偏微分方程式が成り立つというところは前と同じであるが、今度は境界上で外向き法線方向の微分 $\partial u / \partial n$ (これは内向き流速に比例する) が指定される場合に利用します。


START(a,b) segment TO(c,d)... 境界パスの設定コマンド. (a,b) と (c,d) はセグメントの両端を示す座標点. segment には LINE, SPLINE, ARC のいずれかを利用します.

ELEVATION(arg1,[arg2,...]) path 引数の値を縦軸, パスの位置を横軸とする 2次元曲線図を作成します. パスは FROM (X1,Y1) TO (X2,Y2) または ON name で指定します. name には BOUNDARIES セクションで設定したパス名を用います.

NORMAL(vector) 境界に垂直な引数 vector の大きさ (スカラー値) です.

画像のエクスポート

この説明書にもあるように, プロットしたグラフを画像として利用する場合は, 目的の画像でマウスの右ボタンをクリックして EXPORT コマンドを利用します.



図に示すように目的のファイル形式を選択して OK ボタンをクリックします.

時間に依存する問題

FlexPDE は時間を示す変数として“T”や“t”を自動認識します. ここでは先の問題の境界に時間変化する熱容量 C を追加することにします. 時間依存する問題を作成する場合に, 必ず追加すべき項目がありますので, それを次に示します.

- 変数の閾値 THRESHOLD. ただし, INITIAL VALUE セクションで初期値を設定した場合, これは不要です.
- 時間に依存する偏微分方程式
- 時間範囲

TIME <start> TO <finish>

- プロットを作図する時間間隔

```
FOR T=<t1> BY<step> TO <t2>...
```

- 履歴を示すプロットは必要に応じて作成します. HISTORIES セクションを利用します.

```
HISTORY(arg1[,arg2,...]) AT (X1,Y1)[(X2,Y2)...
```

```
TITLE 'Heat flow around an Insulating blob'
COORDINATES cartesian2
VARIABLES
  phi(threshold=0.01) { 閾値を設定します }
DEFINITIONS
  k=1
  c=1 { 熱容量 }
  r=0.5
EQUATIONS
  div(-k*grad(phi))+C*dt(phi)=0 { 熱容量の項を追加します }
BOUNDARIES
  REGION 1 'box'
    START(-1,-1)
    VALUE(phi)=0 LINE TO (1,-1)
    NATURAL(phi)=0 LINE TO (1,1)
    VALUE(phi)=sin(t) LINE TO (-1,1) { 時間の経過とともに変化 }
    NATURAL(phi)=0 LINE TO CLOSE
  REGION 2 'blob'
    k=0.001
    C=0.1 { こちらにも追加します }
    START 'ring' (r,0)
    ARC(CENTER=0,0) ANGLE=360 TO CLOSE
TIME 0 TO 2*pi { 時間範囲 }
PLOTS
  FOR T=pi/2 BY pi/2 TO 2*pi { プロットのタイミング }
    CONTOUR(phi)
    VECTOR(-k*grad(phi))
    ELEVATION(Phi) FROM (0,-1) to (0,1)
HISTORIES { ある座標点における履歴のプロット }
  HISTORY(Phi) AT (0, r/2) (0, r) (0, 3*r/2)
END
```

アニメーション

時間に依存して変化するプロットのアニメーションを作成する機能について説明します。アニメーションの作成には2つの方法があります。その用途に合わせて方法が異なります。


方法 1: FlexPDE の中でアニメーションを再生する

方法 2: ウィンドウズの GIF アニメーションファイルを作成する


FlexPDE 中でのアニメーション

先ほど作成したスクリプトファイルを利用します。画面に表示されているスクリプトファイル名を test-3.pde とします。スクリプトファイルを実行すると、そのプロット結果を保存するファイル test-3.pg5 を自動的に作成します。FlexPDE の中で時間変化する流束の様子をアニメーションする場合は拡張子 .pg5 のファイルを利用します。

1. 画面に上に test-3.pde のファイルを開いたままでもかまいません。File:Open File として test-3.pg5 を選択します。pde ファイルの出力画面と同じファイルを表示します。タブのアイコンだけが異なります。



2. アニメーションを作成するプロットをダブルクリックして最大化表示します。ここではプロット Phi をダブルクリックします。




3. View:Movie と操作します。操作画面上でアニメーションを表示します。描画するフレームの間隔を調整する場合は、View:Frame Delay コマンドを利用します。また、再度アニメーションを実行する前に必ず View:Restart と操作してスタート時点に戻してから Movie コマンドを選択します。

GIF アニメーション

こちらは一般的な GIF アニメーションファイルを作成する方法です。FlexPDE では単に複数枚の連続した png ファイルを作成するだけです。よって、ANIMATION Pro などの市販ソフトでそれらの png ファイルからアニメーションファイルを作成します。手順の 1 と 2 は方法 1 と同じです。

1. 画面に上に test-3.pde のファイルを開いたままでもかまいません。File:Open File として test-3.pg5 を選択します。pde ファイルの出力画面と同じファイルを表示します。タブのアイコンだけが異なります。
2. アニメーションを作成するプロットをダブルクリックして最大化表示します。ここではプロット Phi をダブルクリックします。
3. View:Export Movie と操作します。次のダイアログでフォーマットとして PNG を選択します。



- test-3.pg5 の存在するフォルダに test-301.001 から 004 まで 4 つの png ファイルを作成します。これらのファイルから GIF アニメーションを作成する方法は利用するプログラムによって異なります。

アニメーションの動きをもっと滑らかなものにする場合はスクリプトファイルでプロット間隔を制御する次のコマンド行を編集します。

```
PLOTS
```

```
FOR T=pi/2 BY pi/16 TO 2*pi { 増分を pi/2 から pi/16 に変更 }
```

```
CONTOUR(phi)
```

以上でデモ版の操作説明を終了します。

このデモ版用ガイドブックの作成に際し、下記の書籍を参考にしました。

スタンリー・ファローウ著, 伊理正夫・伊理由美訳「偏微分方程式」, 朝倉書店 (2005)

株式会社ライトストーンでは FlexPDE の基本操作編講習会 (無料) を実施中です。この講習会は購入前または購入後のお客様を対象に、FlexPDE の製品コンセプトと基本的な操作方法をご理解いただくことを目的としています。少人数制で個別のご質問にも対応させていただきます。詳細は弊社のウェブサイトをご覧ください。

株式会社ライトストーン

〒101-0031 東京都千代田区東神田2-5-12 龍角散ビル7F

TEL: 03-3864-5211

FAX: 03-3865-0050

e-mail:tech@lightstone.co.jp

web:www.lightstone.co.jp